
ABSTRACT
I describe algorithms for drawing from distributions using adaptive Markov chain Monte
Carlo (MCMC) methods, introduce a Mata function for performing adaptive MCMC,
amcmc(), and a suite of functions amcmc *() allowing an alternative implementation
of adaptive MCMC. amcmc() and amcmc *() may be used in conjunction with models
set up to work with Mata’s [M-5] moptimize( ) or [M-5] optimize( ), or with standalone functions. To show how the routines might be used in estimation problems, I give
two examples of what Chernozukov and Hong (2003) refer to as Quasi-Bayesian or
Laplace-Type estimators - simulation-based estimators employing MCMC sampling. In
the first example I illustrate basic ideas and show how a simple linear model can be
estimated by simulation. In the next example, I describe simulation-based estimation of
a censored quantile regression model following Powell (1986); the discussion describes
the workings of the Stata command mcmccqreg. I also present an example of how the
routines can be used to draw from distributions without a normalizing constant, and in
Bayesian estimation of a mixed logit model. This discussion introduces the Stata
command bayesmlogit.

Adaptive MCMC in Mata

1

Introduction

Markov chain Monte Carlo (MCMC) methods are a popular and widely-used means
of drawing from probability distributions that are not easily inverted, that have
difficult normalizing constants, or for which a closed form cannot be found. While
often thought of as a collection of methods with primary usefulness in Bayesian
analysis and estimation, MCMC methods can be applied to a wide variety of estimation problems. Chernozukov and Hong (2003), for example, show that MCMC
methods may be applied to many problems of traditional statistical inference and
can be used to estimate a wide class of models - essentially, any statistical model
with a pseudo-quadratic objective function. This class of models encompasses
many common econometric models that have traditionally estimated by maximum likelihood or generalized-methods of moments. This paper describes some
Mata functions for drawing from distributions using a few different types of socalled “adaptive MCMC” algorithms. The Mata implementation of the algorithms
is intended to allow straightforward application to estimation problems.
While their usefulness in drawing from difficult densities is well-known, why
might one wish to employ MCMC methods in estimation? Sometimes maximization of an objective function may be difficult and/or slow, perhaps due to discontinuities or non-concave regions of the objective function, a large parameter space,
and/or difficulty in programming analytic gradients/Hessians. When bootstrapping of standard errors is required, estimation problems are exacerbated because
of the need to re-estimate a model a large number of times. MCMC methods
may provide a more feasible alternative means of estimation in these cases, as estimation based on sampling directly from the joint parameter distribution does not
require optimization and (in principle) provides the desired end result of estimation
anyways - a description of the joint distribution of parameters. MCMC methods
are popular means of implementing Bayesian estimators because they allow one
to avoid calculation of hard-to-calculate normalizing constants that often appear
in posterior distributions. Unlike extrema-based estimation, Bayesian estimators
do not rely on asymptotic results, and thus are useful in small-sample estimation problems or other cases in which the asymptotic distribution of parameters is
difficult to characterize.
In this paper I describe a Mata function, amcmc(), that implements adaptive
or nonadaptive MCMC algorithms and a suit of routines amcmc *() that allow
implementation via a series of structured commands, as one might use Mata functions such as [M-5] moptimize( ) or [M-5] deriv( ). The algorithms implemented
by the Mata routines more or less follow the presentation of Andrieu and Thoms
(2008), who present an accessible overview of the theory and practice of adaptive
MCMC.
In section 2 I provide an intuitive overview of adaptive MCMC algorithms,
while in section 3 I describe how the algorithms are implemented in Mata by
amcmc() or through creation of a structured object via the suite of commands
amcmc *(). In section 4, I describe four applications. In the first, to fix ideas I
show how the routines might be employed in a straightforward parameter estimation problem. In the second I describe how methods can be applied to a more
difficult problem: censored quantile regression. This discussion also introduces
the Stata command mcmccqreg. I then show how routines can be used to sample

from a distribution that lacks a normalizing constant and is hard to invert. In a
final example I apply the methods to Bayesian estimation of a mixed logit model
following Train (2009), and introduce the Stata command bayesmlogit. In section 5, I sketche a basic Mata implementation of an adaptive MCMC algorithm,
in the hope of giving users a template for developing adaptive MCMC algorithms
in more specialized applications. In Section 6 I conclude and offers some sources
for additional reading.

2

An Overview of adaptive MCMC algorithms

At the heart of adaptive MCMC sampling is the Metropolis-Hastings (MH) algorithm. An MH algorithm is built around a target distribution that one wishes to
sample from, π(X), and a proposal distribution q(Y, X).1 If one is mainly interested in applying MCMC in estimation, one may think of π(X) as a conditional
likelihood function, and X can be thought of as a 1 × n row vector of parameters.
A basic MH algorithm is described in table 1.

Basic MH algorithm
1: Initialize start value X = X0 and draws T .
2: Set t = 0 and repeat steps 3-6 while t ≤ T :
3: Draw a candidate Yt from hq(Yt , Xt ).

π(Yt ) q(Yt ,Xt )
4: Compute α(Yt , Xt ) = min π(X
,1
t ) q(Xt ,Yt )
5: Set Xt+1 = Yt with prob. α(Yt , Xt ),
Xt+1 = Xt with prob. 1 − α(Yt , Xt ).
6: Increment t.
Output: The sequence {Xt }Tt=1

i

Table 1: A Metropolis-Hastings algorithm. The proposal distribution is denoted by q(Y, X), while the target distribution is π(X).
α(X, Y ) denotes the draw acceptance probability.
The MH algorithm sketched in table 1 has the property that candidate draws Yt
increasing the value of the target distribution π(X) are always accepted, whereas
candidate draws that produce lower values of the target distribution are only accepted with probability α. Under fairly general conditions, the draws X1 , X2 , ..., XT
converge to draws from the target distribution π(X); see Chib and Greenberg
(1995) for proofs. One can see the convenience the algorithm provides in drawing
from densities of the form π(X) = π ′ (X)/K, where K is some perhaps difficultto-calculate normalizing constant. Computation of K is unnecessary, as it cancels
out of the ratio π(X)
. The proposal distribution q(Y, X) is where the “Markov
π(Y )
chain” part of “Markov chain Monte Carlo” comes in, and is what distinguishes
MCMC algorithms from more general acceptance-rejection Monte Carlo sampling,
as it is through this function that candidate draws depend upon previous draws.
MCMC algorithms are simple and flexible, and are therefore applicable to a
wide variety of problems, but implementation can be challenging, mainly because
1. For ease of comparison, wherever possible I follow the notation of Andrieu and Thoms (2008).

finding an appropriate proposal distribution q(Y, X) can be hard. If q(Y, X) is
chosen poorly, coverage of the target distribution π(X) may be poor. This is where
adaptive MCMC methods come into play, as they provide a means of “tuning” the
proposal distribution. As an adaptive MCMC algorithm proceeds, information
about acceptance rates of previous draws is collected and embodied in some set
of tuning parameters θ. Slow or non-convergence of an algorithm like that in
table 1 is often caused by acceptance of too few or too many candidate draws
- if the algorithm accepts too few candidate draws, candidates are too far away
from regions of the support of the distribution where π(X) is large, while if too
many candidates are accepted, candidates occupy an area of the support of the
distribution clustered closely about a large value of π(X). Accordingly, if the
acceptance rate is too low, the typical tuning mechanism contracts the search
range, while if the acceptance rate is too high, the range is expanded. As a practical
matter, one augments the proposal distribution with the tuning parameter(s) θ, so
that the proposal distribution is something like q(Y, X) = q(Y, X, θ). A description
of such an algorithm appears in table 2.
The algorithm described in table 2 also relies on a simplification of the basic
MCMC algorithm presented in table 1 which results when a symmetric proposal
distribution is used, so that q(Y, X, θ) = q(X, Y, θ). With a symmetric proposal
distribution - the (multivariate) normal distribution being a prominent example
- the proposal distribution drops out of the calculation of the acceptance probability in step 4 of
resulting in the simplified acceptance probability
h the algorithm,
i
α(Y, Xt ) = min

π(Y )
,1
π(Xt )

. All of the Mata routines discussed in this paper use a

multivariate normal density for a proposal distribution.

Adaptive MH algorithm (with symmetric q)
1: Initialize start value X = X0 , draws T , and tuning parameter(s) θ0 .
2: Set t = 0 and repeat steps 3-7 while t ≤ T :
3: Draw a candidate Yt from hq(Yt , Xt ,iθt ).
π(Yt )
,1
4: Compute α(Yt , Xt ) = min π(X
t)
5: Set Xt+1 = Yt with prob. α(Yt , Xt ),
Xt+1 = Xt with prob. 1 − α(Yt , Xt ).
6: Update θt+1 = f (θt , X0 , X1 , X2 , ..., Xt ).
7: Increment t.
Output: The sequence {Xt }Tt=1

Table 2: Overview of an adaptive Metropolis-Hastings algorithm with tuning
and a symmetric proposal distribution.
There is an important theoretical problem with an adaptive MCMC algorithm
like that in table 2. Tuning the proposal distribution results in “loss of π as an
invariant distribution of the process {Xt }” (Andrieu and Thoms 2008, p. 345) if it
is not done carefully. The act of tuning the proposal distribution alters the longrun behavior of the algorithm, so that it is no longer producing the sought-after
draws from the target distribution π(X). A solution to this problem is to tune
the proposal distribution for some burn-in period and then stop tuning so that the
proposal distribution is stationary. Another solution is to set up the algorithm so

Matthew J. Baker
that tuning eventually recedes from the algorithm. The latter approach is referred
to as vanishing or diminishing adaptation (Andrieu and Thoms 2008; Rosenthal
2011). With vanishing adaptation, if the algorithm runs for a sufficient number of
iterations, the proposal distribution stabilizes while also (hopefully) being tuned
to provide good coverage of the target distribution. The Mata functions presented
in this paper are built to work with vanishing adaptation, but can also be set up
so that no adaptation of the proposal distribution occurs.

2.1

Adaptive MCMC with vanishing adaptation

A necessary prelude to discussion of implementation of vanishing adaptation is a
discussion of how frequently candidate draws should be accepted by an MCMC
algorithm. Ideally, the acceptance rate should be such that good coverage of
the target distribution is achieved with the smallest possible number of draws.
Rosenthal (2011) contains an accessible treatment on optimal acceptance rates in
adaptive MCMC algorithms and a summary of the main ideas and results. At the
risk of oversimplifying, some guidelines are as follows. For univariate distributions
the optimal acceptance rate is about .44, and as the dimension of π(X) increases to
infinity, the optimal acceptance rate converges to .234. Rosenthal (2011) points out
that moderate departure from these rates is not likely to greatly damage algorithm
performance, and that in many cases for distributions even with relatively small
dimension (i.e, d ≥ 5) the optimal acceptance rate is close to the asymptotic bound
of .234. Given a targeted acceptance rate α∗ (presumably in or close to the range
[.234, .44]) table 3 describes an adaptive MCMC algorithm which tunes towards
an acceptance rate of some given value of α∗ as it proceeds.

Adaptive MCMC algorithm with normal proposal and vanishing adaptation
1: Set starting values X0 , µ0 , Σ0 , λ0 , α∗ , δ (δ > 0), and draws T .
2: Set t = 0 and repeat steps 3-10 while t ≤ T :
3: Draw a candidate Yt ∼ M V
h N (Xt , λit Σt ).
π(Yt )
,1
4: Compute α(Yt , Xt ) = min π(X
t)
5: Set Xt+1 = Yt with prob. α(Yt , Xt ),
Xt+1 = Xt with prob. 1 − α(Yt , Xt ).
1
6: Compute weighting parameter γt = (1+t)
δ.
∗
7: Update λt+1 = exp [γt (α(Yt , Xt ) − α )] λt .
8: Update µt+1 = µt + γt (X

 t+1 − µt )
9: Update Σt+1 = Σt + γt (Xt+1 − µt ) (Xt+1 − µt )′ − Σt
10: Increment t.
Output: The sequence {Xt }Tt=1

Table 3: Overview of an adaptive Metropolis-Hastings algorithm with a multivariate normal proposal distribution and a specific tuning mechanism.
Table 3 is a fairly complete description of how an adaptive MCMC algorithm
might be implemented (and how the Mata functions presented in section 3 actually
operate). In step 1, the algorithm starts with initial value X0 , an initial variancecovariance matrix for proposals, Σ0 , an initial value of a scaling parameter λ0 , and

a targeted acceptance rate α∗ . The algorithm also requires a value for what can
be thought of as an averaging or damping parameter, δ, controlling how quickly
1
the impact of the tuning mechanism decays through the parameter γt = (1+t)
δ,
calculated in step 6. For large values of δ adaptation ceases quickly as γ more
rapidly approaches zero, while for values of δ close to zero, adaptation occurs
more slowly and the algorithm uses more information about past draws in tuning
proposals. The Mata routines presented below allow the user to specify such a δ
parameter in an implementation of the algorithm.2 In steps 8 and 9, the algorithm
updates the mean and covariance matrix of the proposal distribution according
to the weighting parameter γt , and since γt eventually decays to zero, updating
ceases and the algorithm eventually carries on with stable proposal distribution
characterized by λt+1 = λt , µt+1 = µt , and Σt+1 = Σt .
If a researcher wished to write his or her own adaptive MCMC routine, it
bears mentioning that the choice of the weighting scheme embodied in γ and δ on
table 3 is one place where there is room for extension. Andrieu and Thoms (2008)
describe some other possibilities for adaptation, including stochastic schemes or
weighting functions that themselves adapt as the algorithm continues. In fact, as
described by Andrieu and Thoms (2008, p.356), virtually anything goes with the
tuning process, provided that the sequence γt satisfies the following properties:
∞
X

γt = ∞,

t

∞
X

γt1+ρ < ∞; ρ > 0.

t

These conditions are satisfied by the weighting parameter used in the adaptive
P∞
algorithm on table 3 so long as δ ∈ (0, 1), as under these circumstances,
γt
t
diverges, but a sufficiently large value of ρ can always be found be found that
forces the series

h

1
(1+t)δ

i1+ρ

to converge.

A last detail to address is how to initialize the value of the scaling parameter
λ at the start of the algorithm. According to Andrieu and Thoms (2008, p.359),
theory suggests a good place to start with the scaling parameter is λ ≈ 2.382 /d,
where d is the dimension of the target distribution. The Mata routines presented
below all use this value as a starting point, with one exception.
There are many variations on the basic theme of the algorithm presented in
table 3. One possibility is one-at-a-time, sequential sampling of values from the
distribution, which produces a “metropolis-within-gibbs” type sampler. Another
possibility is to work halfway in between the “global” sampling algorithm of table 3
and sequential sampling, creating what might be labeled a block adaptive MCMC
sampler.3 In the author’s experience, metropolis-within-Gibbs samplers or block
samplers are often useful in situations in which variables are scaled very differently
or in situations where the researcher might not have a lot of intuition about starting
values.
2. One might prefer that this value be as close to its upper bound as possible, so as to reduce the
impact of tuning quickly; the tradeoff is that the proposal distribution may not be as well-adapted
in this case.
3. I follow the convention of referring to a sequential sampler as a “metropolis-within-gibbs” sampler,
even though many find this terminology misleading; see Geyer (2011, p.28-29). What I refer to as
a “block” sampler, some might call a “block-gibbs” sampler.

3
3.1

Adaptive MCMC in Mata
A Mata function

Syntax
The first Mata implementation of the algorithms described in section 2 is through
the mata function amcmc(). amcmc() uses different types of adaptive MCMC samplers based upon user-provided information. In addition to describing details of
sampling (specification of draws, weighting parameters, and acceptance rates), the
user can also specify if sampling is to proceed all at once (“globally”), in blocks, or
sequentially. The user can also set up amcmc() to work with a “stand-alone” distribution or with an objective function previously set up to work with moptimize()
or optimize(). Syntax is as follows:

real matrix amcmc(string rowvector alginfo,
pointer(real scalar function) scalar lnf(),
real rowvector xinit, real matrix Vinit,
real scalar draws, real scalar burn,
real scalar delta, real scalar aopt,
transmorphic arate, transmorphic vals,
transmorphic lambda, real matrix blocks
| transmorphic M, string scalar noisy)
Description
If the dimension of the target probability distribution (or the parameter vector, as
the case may be) is characterized as a 1 × c row vector, amcmc() returns a matrix
of draws from the distribution organized in c columns and r = draws − burn rows,
so each row of the returned matrix can be thought of as a draw from the target
distribution lnf(). Additional information about the draws are collected in three
arguments overwritten by amcmc(): arate, vals, and lam, which contain actual
acceptance rate(s), the log-value of the target distribution at each draw, and λ,
the proposal scaling parameter(s). In the case in which a metropolis-within-gibbs
sampler or a block sampler is used (more on this to follow), lam is returned as a
row vector equal in length to the dimension of the distribution or the number of
blocks, as is arate.
Information about how to go about drawing from the target distribution, and
how the distribution has been programmed is passed to the command as a sequence of strings in the (string) row vector alginfo. This row vector can contain
information about whether sampling is to be sequential (mwg), in blocks (block),
or global (global). In the event that the user is interested in applying amcmc()
to a model statement constructed with moptimize() or optimize(), information
on this, and the type of evaluator function used with the model, should also be
contained in alginfo. Target distribution information can be either standalone,

8

Adaptive MCMC in Mata
moptimize, or optimize. Information on evaluator type can also be of any sort
(i.e., d0, v0, etc.)4 . A final option that can be passed along as part of alginfo is
the key fast, which will execute the adaptive mcmc algorithm more speedily but
less exactly. In the remarks about syntax, some examples as to what alginfo might
look like are described.
The second argument of amcmc(), lnf, is a (pointer to) the target distribution,
which must be written in log form. xinit and Vinit are conformable initial values
for the routine and an initial variance-covariance matrix for the proposal distribution. The scalar draws and burn tell the routine how many draws to make from
the distribution and how many of these draws are to be discarded as an initial
burn-in period. delta is a string scalar which describes how adaptation is to occur,
while aopt is the desired acceptance rate; see section 2.1.
The real matrix blocks contains information about how amcmc() should proceed
if the user wishes to draw from the function in blocks. If one does not wish to
draw in blocks, one simply passes a missing value for this argument. If the user
provides an argument here, but does not specify block as part of alginfo, sampling
will not occur in blocks.
If the user is drawing from a function constructed with a prespecified model
command written to work with either moptimize or optimize, this model statement is passed to amcmc() via the optional M argument. As described below,
this argument can also have other uses; for example, passing up to ten additional
explanatory variables to amcmc().
The final option is noisy, and if the user specifies noisy="noisy", amcmc() will
produce feedback on drawing as the algorithm executes. A dot is produced every
time the evaluation function lnf is called (not every time a “draw” is completed,
as the latter is taken by amcmc() to mean a complete run through the routine).
Thus, in cases in which a block sampler or a metropolis-within-gibbs style sampler
are used, a draw is deemed to have occurred when all the blocks or variables have
been drawn once. Every fifty evaluations, the value of the target distribution is
reported.

Remarks
It is helpful to have a few examples of how information about the draws to be
conducted can be passed to the amcmc() function through the first argument,
alginfo. This is described in table 4:

Sampling information
Model definition
Evaluator type
Other information

mwg,global,block
moptimize,optimize,standalone
d*,q*,e*,g*,v*
fast

Table 4: Options for using amcmc(), passed in the argument alginfo.
One can select any item from each of the rows on table 4 and pass it to amcmc()
4. The routine will not, however, work with type lnf evaluators

as part of alginfo. For example, if one is trying to draw from a function that was
written up as type d2 evaluator to work with moptimize, and the user wished to
use a global sampler, he or she might specify:
alginfo="moptimize","d2","global"
Order doesn’t matter, so the user could also specify:
alginfo="d0","moptimize","global"
If the user had a stand alone function and wished to do metropolis-within-gibbs
style sampling from this function, he or she would specify:
alginfo="standalone","mwg"
Or even just alginfo="mwg", as if no model statement is submitted, amcmc()
will assume that the function is “stand-alone.” The final option that the user
might specify is the “fast” option, which works by tacking on the string fast
to alginfo. This option is designed for situations in which the user wishes to
sample globally or in blocks, but has a problem with large dimension. Since the
global and block samplers use Cholesky decomposition of the proposal covariance
matrix, large problems may be time-consuming. The “fast” option circumvents
the potential slowdown in speed by working with just the diagonal elements of
the proposal covariance matrix, so Cholesky decomposition can be avoided. One
should, however, exercise caution in using this option, and should probably apply
it only when the user can be reasonably certain that distribution variables are
independent.5
The row vector xinit contains an initial value for the draws, while Vinit is an
initial variance covariance matrix that may just be a conformable identity matrix.
If, however, Vinit is a row vector, amcmc() will interpret this as the diagonal of a
variance matrix, with zero off-diagonal entries.
While the user-specified scalar delta controls how rapidly adaptation vanishes,
the user may also specify delta=., and amcmc() will then assume that the user does
not want any adaptation to go on, but instead wishes to draw from the invariant
proposal distribution with mean xinit and covariance matrix Vinit. In this case,
the user must supply values of lambda to describe to the algorithm how to scale
draws from the proposal distribution. The idea in constructing the code this way is
to allow users to run the adaptive algorithm for awhile, and once it has converged,
allow the user to switch to an algorithm using an invariant proposal distribution.
If a global sampler is used, only one value of lambda is required, otherwise, lambda
must be conformable with the sampler, so, if the option mwg is being used, the
dimension of lambda must match the dimension of the target distribution, while if
the option block is used, lambda must contain as many entries as the number of
blocks.
Whether one wishes to do metropolis-within-gibbs sampling, block sampling, or
global sampling, the routine requires the same set of input information (although
the overwritten values lam and arate differ slightly) with one exception. When
sampling in block form, amcmc() requires that a matrix be provided in block, in
5. In fact, I included this option in the hopes that users might try it out and see for what sorts of
problems it does, and does not, work well, if any.

which the number of rows is equal to the number of sampling groups, and those
values which are to be drawn together have ones in the appropriate positions and
zeros elsewhere. So, for example, if one wished to draw from a five-dimensional
distribution, and wished to draw values for the first three arguments together, and
then arguments four and five together, one would set up a matrix B as follows:
B=



1
0

1
0

1
0

0
1

0
1



One can also pass as a block matrix an identity matrix:






B=

1
0
0
0
0

0
1
0
0
0

0
0
1
0
0

0
0
0
1
0

0
0
0
0
1







The reader might suspect that this would result in the same sort of algorithm
obtained by specifying alginfo="mwg", but this is not the case, because after each
draw, the block algorithm updates the entire mean proposal vector and covariance
matrix, so information on each draw is used in preparing for the next.6 While not
the intended use of the block-sampling algorithm, by leaving a column all zeroes
in the matrix B, the corresponding value of the parameter will never be drawn.
This is a quick, albeit not particularly efficient, way of constraining parameters at
particular values during the drawing process.
The argument M of amcmc() can contain a previously assembled model statement, or it can also be used to pass additional arguments of a function to the
routine.7 . As an example, if the user has written a function to be sampled from
that has three arguments, such as lnf(x,Y,Z), the user would simply specify the
standalone option in the variable alginfo, assemble the additional arguments into
a pointer, and then pass this information to amcmc(). In this instance, M might be
constructed in Mata as follows:

M=J(2,1,NULL)
M[1,1]=&Y
M[2,1]=&Z
M can then be passed to amcmc(), which will use Y and Z (in order) in evaluating
lnf(x,Y,Z). As will be made clear in the examples, this usage of pointers can be
handy when amcmc() is used as part of a larger algorithm, as one can continually
change Y and Z without actually having to explicitly declare that Y and Z have
changed as the algorithm executes.
6. Using amcmc() in this way is akin to what Andrieu and Thoms (2008, p.360) describe as an adaptive
MCMC algorithm with “componentwise adaptive scaling.”
7. But not both; the assumption is that any arguments have already been built into the model
statement if a previously-constructed model is used

Matthew J. Baker

3.2

Adaptive MCMC via a structure

Syntax
Another alternative which has some advantages in certain situations, particularly
when one wishes to do adaptive MCMC as one step in a larger sampling problem,
is to set up an adaptive MCMC sampling problem using the set of commands
amcmc *(). The user first opens a problem using the amcmc init() command, and
then fills in the details of the drawing procedure. The following commands can be
used to set up an adaptive MCMC problem, with the arguments corresponding to
those described in section 3.1:

A = amcmc init()
amcmc lnf(A,pointer (real scalar function) scalar f)
amcmc args(A,pointer matrix Z)
amcmc xinit(A,real rowvector xinit)
amcmc Vinit(A,real matrix Vinit)
amcmc aopt(A,real scalar aopt)
amcmc blocks(A,real matrix blocks)
amcmc model(A,transmorphic M)
amcmc noisy(A,string scalar noisy)
amcmc alginfo(A,string rowvector alginfo)
amcmc damper(A,real scalar delta)
amcmc lambdas(A,real rowvector lambda)
amcmc draws(A,real scalar draws)
amcmc burn(A,real scalar burn)
Once a problem has been specified, a run can be initiated via the command:

amcmc draw(A)
Results can be accessed via a series of commands of the form:

amcmc results *(A)
where * in the above can be any of the following: vals, arate, passes,
totaldraws, acceptances, propmean, propvar. Additionally, the user can recover
his/her initial specifications by using * = draws, aopt, alginfo, noisy, blocks,
damper, xinit, Vinit, or lambda. An additional function amcmc results lastdraw()
produces only the value of the last draw. Two other functions are handy when one
is executing an adaptive MCMC draw as part of a larger algorithm. These are:

amcmc append(A,string scalar append)

amcmc reeval(A,string scalar reeval)
The function amcmc append() allows the user to describe whether or not results
should be overwritten, by specifying append ="overwrite". In this case, only the
results of the most recent draw(s) are kept. This can be useful when doing an
analysis where nuisance parameters of a model are being drawn, and storing all
the previous draws would tax the memory and impact the speed of the algorithm’s
operation. The function amcmc reeval() allows the user to indicate whether or not
the target distribution should be reevaluated at the last draw before a proposed
value is tried by specifying reeval ="reeval". When the draw is part of a larger
algorithm, some of the arguments of the target distribution might have changed
as the larger algorithm proceeds. In these cases, the target distribution needs to
be reevaluated at the new argument values and last previous draw to function
correctly. If the user sets reeval to anything else, it is assumed that nothing has
changed, and that the value of the target distribution has not changed between
draws.

Remarks
Implicit in some of the information accessible with amcmc results *() are some
hints as to why a user might prefer to use a problem statement to attack an
adaptive MCMC problem instead of just applying the mata function amcmc(). A
chief usefulness stems from the ease with which one may stop, restart, and append a run within Mata’s structure environment. In this way a user can perform
adaptive MCMC as part of a larger algorithm; the structure allows information
about past adaptation and runs to be easily retained as the algorithm proceeds,
while at the same time arguments of the algorithm can be easily modified. In
the model statement syntax, information about the number of times a given problem has been initiated is retrievable via the command amcmc results passes(A),
while one can also view the acceptance history of an entire run by accessing
amcmc results acceptances(A).
Given the initialization of an adaptive MCMC problem A, one can run the
amcmc draw() command sequentially and results will be appended to previous
results. Accordingly, the burn period is only active the first time the command is
executed. Thereafter, it is assumed that the user wishes to retain all drawn values.
As mentioned above, whether or not the user wishes to retain all the information
about previous draws or not is controlled through the function amcmc append().
When a user specifies append ="overwrite", so that only the draws of the last run
are saved, the routine still builds in all information about adaptation contained in
the entire drawing history.
When a user initializes an adaptive MCMC problem via amcmc init(), some
defaults are provided unless overwritten by the user. The number of draws is set to
one, the burn period is set to zero, the target distribution is assumed to be standalone, and the acceptance rate is set to .234, and, as previously mentioned, results
are appended to previous results if multiple passes are made, and it is assumed
that the function does not need to be reevaluated at the last value before drawing
a new proposal.

Further description can be found in the help files, accessible by typing help
mata amcmc() or help mf amcmc at Stata’s command prompt.

4
4.1

Examples
Parameter estimation

I start with an example of application of adaptive MCMC to a simple estimation
problem. Suppose that I have already programmed a likelihood function for use
with moptimize() in Mata, but wish to try another means of estimating parameters, perhaps because I have found that maximization of the likelihood function
is taking too long or presents other difficulties, or because I am worried about
small-sample properties of the estimators. I decide to try to estimate the model
by drawing directly from the conditional distribution of parameters. The ideas
derive from Bayes’ rule and the usual principles of Bayesian estimation, but they
can be applied to virtually any maximum likelihood problem.8 Via Bayes’ rule,
the distribution of parameters conditional on the data can be written as:
p(β|X) =

p(X|β)p(β)
p(X|β)p(β)
= R
p(X)
p(X|β)p(β)dβ

(1)

If one has no prior information about parameter values, one can take p(β) - the
prior distribution of parameters - to be (improper) uniform over the support of
the parameters. As this renders p(β) constant, one then obtains the posterior
parameter distribution as:
p(β|X) ∝ p(X|β)
(2)
So, according to equation (2), one might interpret a likelihood function as the distribution of parameters conditional on data (up to a constant of proportionality).
The conditional mean of parameter values is then:
E[β|X] =

Z

βp(β|X)dβ

(3)

E[β|X] can be estimated by simulating the right-hand side of equation (4) via S
draws from the conditional distribution p(β|X):
E[β|X] ≈=

S
1X
β(s)
S

(4)

s=1

These simulations can be used to characterize higher-order moments of the parameter distribution as well. I shall follow the nomenclature adopted by Chernozukov and Hong
(2003) and refer to estimators so obtained as LTEs (Laplace-type estimators) or
QBEs (Quasi-Bayesian estimators).
Returning to the example, suppose I have posited a simple linear model with
log-likelihood function:
ln L ∝

(y − Xβ)′ (y − Xβ)
n
− ln σ 2
2σ 2
2

8. And in fact a much wider variety of problems; see Chernozukov and Hong (2003)

14

Adaptive MCMC in Mata
For purposes of comparison, in the following code snippet I take this simple model
and fit it to some data using a type d0 evaluator and Mata’s moptimize command.
One subtlety of the code is that I have coded the variance in exponentiated form.
This is done so that when amcmc() is applied to the problem, the objective function
is consistent with the multivariate normal proposal distribution, which requires
that parameters have support (−∞, ∞).9 The following code develops the model
statement and estimates the model via maximum likelihood:
. clear all
. sysuse auto
(1978 Automobile Data)
. mata:
mata (type end to exit)
: function lregeval(M,todo,b,crit,s,H)
> {
> real colvector p1, p2
> real colvector y1
>
p1=moptimize_util_xb(M,b,1)
>
p2=moptimize_util_xb(M,b,2)
>
y1=moptimize_util_depvar(M,1)
>
crit=-(y1:-p1)´(y1:-p1)/(2*exp(p2))- ///
>
rows(y1)/2*p2
> }
note: argument todo unused
note: argument s unused
note: argument H unused
:
: M=moptimize_init()
: moptimize_init_evaluator(M,&lregeval())
: moptimize_init_evaluatortype(M,"d0")
: moptimize_init_depvar(M,1,"mpg")
: moptimize_init_eq_indepvars(M,1,"price weight displacement")
: moptimize_init_eq_indepvars(M,2,"")
: moptimize(M)
initial:
f(p) =
-18004
alternative:
f(p) = -10466.142
rescale:
f(p) = -298.60453
rescale eq:
f(p) = -189.39334
Iteration 0:
f(p) = -189.39334 (not
Iteration 1:
f(p) = -172.06827 (not
Iteration 2:
f(p) = -162.08289 (not
Iteration 3:
f(p) = -156.61458 (not
Iteration 4:
f(p) = -143.6168
Iteration 5:
f(p) = -128.64046
Iteration 6:
f(p) = -127.05628
Iteration 7:
f(p) = -127.05447
Iteration 8:
f(p) = -127.05447
: moptimize_result_display(M)

concave)
concave)
concave)
concave)

Number of obs
mpg

Coef.

Std. Err.

z

P>|z|

=

74

[95% Conf. Interval]

9. Another, less efficient way of dealing with parameters having restricted supports is to program the
distribution so that it returns a missing value whenever a draw lands outside of the appropriate
range.

eq1
price
weight
displacement
_cons

-.0000966
-.0063909
.0054824
40.10848

.0001591
.0011759
.0096492
1.974221

-0.61
-5.44
0.57
20.32

0.544
0.000
0.570
0.000

-.0004085
-.0086956
-.0134296
36.23907

.0002153
-.0040862
.0243945
43.97788

2.433905

.164399

14.80

0.000

2.111688

2.756121

eq2
_cons
: end

I now estimate model parameters via simulation by treating the likelihood function like the parameters’ conditional distribution. I first start with a metropoliswithin-gibbs sequential sampler to obtain 10000 draws for each parameter value,
discarding the first 20 draws as a burn-in period. I start with this sampler because
it is usually a relatively safe choice when there is little information on starting
points, which I am pretending are unavailable. I set the initial values used by
the sampler to zero, and use an identity matrix as an initial covariance matrix
for proposals. I choose a value of delta=2/3, which allows a fairly conservative
amount of adaptation to occur, and a desired acceptance rate of 0.4:10
. set seed 8675309
. mata:
mata (type end to exit)
: alginfo="moptimize","d0","mwg"
: b_mwg=amcmc(alginfo,&lregeval(),J(1,5,0),
>
I(5),10000,50,2/3,.4,
>
arate=.,vals=.,lambda=.,.,M)
:
: st_matrix("b_mwg",mean(b_mwg))
: st_matrix("V_mwg",variance(b_mwg))
: end
. local names eq1:price eq1:weight eq1:displacement eq1:_cons eq2:_cons
. mat colnames b_mwg=`names´
. mat colnames V_mwg=`names´
. mat rownames V_mwg=`names´
. ereturn post b_mwg V_mwg
. ereturn display
Coef.

Std. Err.

z

P>|z|

[95% Conf. Interval]

eq1
price
weight
displacement
_cons

-.0001322
-.0057418
.00218
39.00328

.0001714
.0018016
.0125846
3.095009

-0.77
-3.19
0.17
12.60

0.440
0.001
0.862
0.000

-.0004681
-.009273
-.0224854
32.93717

.0002036
-.0022107
.0268454
45.06939

2.518081

.2071915

12.15

0.000

2.111993

2.924169

eq2
_cons

In spite of the fact that the algorithm was not allowed a very long burn in time,
10. A comment about what might seem to be a relatively short burn-in period: I have selected this
burn-in period to be short enough so that one can see the convergence behavior of the algorithm.

the simulation-based parameter estimates are close to those obtained by maximum
likelihood.11 How frequently were draws of each parameter accepted, and how
close is the algorithm working around the maximum value of the function? This
information is returned as the overwritten arguments arate and vals:
. mata:
mata (type end to exit)
: arate´
1
1
2
3
4
5

.3806030151
.3807035176
.3870351759
.4020100503
.3951758794

: max(vals),mean(vals)
1
1

-127.1097198

2
-130.2193494

: end

The sampler finds and operates close to the maximum value of the (log) likelihood
(which was -127.05), and the acceptance rates of the draws are very close to the
desired acceptance rate of .4. To get a sense as to what the distribution of the
parameters looks like, I pass the information about parameter draws to Stata and
form visual pictures of results. The code below accomplishes this and creates two
panels of graphs: one which shows the distribution of parameters (figure 1), and
another which shows how parameter draws and the value of the function evolved
as the algorithm moved (figure 2).
. preserve
. clear
. getmata (b_mwg*)=b_mwg
. getmata vals=vals
. gen t=_n
. local graphs
. local tgraphs
. forvalues i=1/5 {
2.
quietly {
3.
histogram b_mwg`i´, saving(b_mwg`i´, replace) nodraw
4.
twoway line b_mwg`i´ t, saving(bt_mwg`i´, replace) nodraw
5.
}
6.
local graphs "`graphs´ b_mwg`i´.gph"
7.
local tgraphs "`tgraphs´ bt_mwg`i´.gph"
8.
}
.
histogram vals, saving(vals,replace) nodraw
(bin=39, start=-183.40158, width=1.4433811)
(file vals.gph saved)

11. One issue that might be raised at this point is whether or not it is appropriate to summarize
the results in usual Stata format like this. Implicit in the assumption that this is okay is that
the parameters are collectively normally distributed. Whether or not this is true in more general
problems requires careful thought.

Figure 1: The distribution of the parameters after an MCMC run.

.
twoway line vals t, saving(vals_t,replace) nodraw
(file vals_t.gph saved)
.
. graph combine `graphs´ vals.gph
. graph export vals_mwg.eps, replace
(file vals_mwg.eps written in EPS format)
. graph combine `tgraphs´ vals_t.gph
. graph export valst_mwg.eps, replace
(file valst_mwg.eps written in EPS format)
. restore

Figure 1 is comprised of histograms for each parameter, with the last panel the
histogram of the log-likelihood. Parameters seems to be approximately normally
distributed (with a few blips), excepting the first few draws, and they are also
centered around parameter values obtained via maximum likelihood. Figure 2
shows how the drawn values for parameters and the value of the objective function
evolved as the algorithm proceeded. From figure 2, one can see that after a few
iterations, the algorithm settles down to drawing from an appropriate range. The
draws are also clearly autocorrelated, and this autocorrelation is a general property
of any MCMC algorithm, adaptive or not. For this reason, when applying MCMC
algorithms in practice, it is sometimes beneficial to thin out the draws by keeping,
say, only every fifth or tenth draw or jumble draws. Some sources describing
additional tips for analyzing and presenting the results of a MCMC run appear in
the conclusion.
To illustrate use of a global sampler, and also some of the problems one might

Figure 2: A look at the estimates

encounter in doing an MCMC based analysis, I now apply a global sampler to
the problem so that all parameter values are drawn simultaneously. The following
snippet of code shows the results of a run of 12000 draws with a burn-in period of
2000:
. set seed 8675309
. mata:
mata (type end to exit)
: alginfo="global","d0","moptimize"
: b_glo=amcmc(alginfo,&lregeval(),J(1,5,0),
>
I(5),12000,2000,2/3,.4,
>
arate=.,vals=.,lambda=.,.,M)
:
: st_matrix("b_glo",mean(b_glo))
: st_matrix("V_glo",variance(b_glo))
: end
. local names eq1:price eq1:weight eq1:displacement eq1:_cons eq2:_cons
. mat colnames b_glo=`names´
. mat colnames V_glo=`names´
. mat rownames V_glo=`names´
. ereturn post b_glo V_glo
. ereturn display
Coef.
eq1

Std. Err.

z

P>|z|

[95% Conf. Interval]

Figure 3: Distribution of parameters after a global MCMC run that is slow to converge.

price
weight
displacement
_cons

-.0004614
.013056
-.1798405
15.16227

.0019104
.0232029
.3163187
20.84814

-0.24
0.56
-0.57
0.73

0.809
0.574
0.570
0.467

-.0042057
-.0324209
-.7998138
-25.69933

.0032829
.0585328
.4401328
56.02388

4.017743

1.880032

2.14

0.033

.3329483

7.702537

eq2
_cons

One can see from these results that the algorithm has not so quickly found an
appropriate range of values for parameter values. Figures 3 and 4 give an indication
as to why; the algorithm spends considerable time stuck away from the maximal
function value. The biggest lesson of figures 3 and 4 is that the algorithm was
not allowed to burn in for a significantly long period of time for the global MCMC
algorithm to work correctly. While the parameter values eventually settle down
closer to their “true” values, it has taken the algorithm upwards of 6000 draws to
find the right range. In fact, it looks as though the algorithm settled into a stable
range for draws 2000-6000 or so, but then once again experienced a jump to the
correct stable range, a phenomenon known as “psuedo-convergence” (Geyer 2011).
This behavior is also responsible for the multimodal appearance of the histograms
on figure 3.
While my intent here is to simply illustrate how the mata function amcmc()
works, the example also illustrates a few points about what can happen when
one is not careful in specifying adjustment parameters and allowing an adaptive
MCMC algorithm to run long enough in a given estimation problem. One may
get bad results without knowing it, as the case would be if the global algorithm

Figure 4: Characteristics of draws after a global MCMC run

had only been allowed to run for 5000 iterations. This sometimes happens if poor
starting values are mixed with parameters that have very different magnitudes, as
is the case with the constant in the initial model relative to the other parameters.
One can see from inspecting figure 3 is that the constant did not find its correct
range until just after 6000 draws, and this is likely the cause of the difficulty.
This discussion motivates using amcmc() in steps, where a slower but relatively
robust sampler (a metropolis-within-gibbs sampler in this case) is used to orient
parameters close to their correct range before a global sampler is used, as is done
in the following snippet:
. mata:
mata (type end to exit)
: alginfo="mwg","d0","moptimize"
: b_start=amcmc(alginfo,&lregeval(),J(1,5,0),
>
I(5),5*1000,5*100,2/3,.4,
>
arate=.,vals=.,lambda=.,.,M)
: alginfo="global","d0","moptimize"
: b_glo2=amcmc(alginfo,&lregeval(),mean(b_start),
>
variance(b_start),11000,1000,2/3,.4,
>
arate=.,vals=.,lambda=.,.,M)
: st_matrix("b_glo2",mean(b_glo2))
: st_matrix("V_glo2",variance(b_glo2))
: end
.
. local names eq1:price eq1:weight eq1:displacement eq1:_cons eq2:_cons

. mat colnames b_glo2=`names´
. mat colnames V_glo2=`names´
. mat rownames V_glo2=`names´
. ereturn post b_glo2 V_glo2
. ereturn display
Coef.

Std. Err.

z

P>|z|

[95% Conf. Interval]

eq1
price
weight
displacement
_cons

-.0001059
-.0063727
.0056462
40.10216

.0001584
.0012014
.0099215
1.912111

-0.67
-5.30
0.57
20.97

0.504
0.000
0.569
0.000

-.0004164
-.0087275
-.0137997
36.35449

.0002046
-.0040179
.025092
43.84982

2.480892

.1665249

14.90

0.000

2.15451

2.807275

eq2
_cons

Thus, one is free to begin by drawing parameters that are scaled very differently
either alone or in blocks until the algorithm finds it footing, and then proceed with
a global algorithm.
Another alternative is once again beginning with a metropolis-within-gibbs
sampler to characterize the distribution of the parameters, and, once this is done
sufficiently well, run the algorithm without adaptation so one is using an invariant
proposal distribution and a regular MCMC algorithm. After an initial run with
the "mwg" option, I submit the mean and variance of results to the global sampler
with no adaptation parameter; passing a value of missing (.) for delta. Since I am
not passing any information to amcmc() about how to go about adaptation in this
case, it requires a value for lambda to be submitted, so I choose λ = 2.382 /n.12
Finally, I also submit a missing value for aopt. Since no adaptation is occurring,
aopt is not used by the algorithm.
. mata:
mata (type end to exit)
: alginfo="mwg","d0","moptimize"
: b_start=amcmc(alginfo,&lregeval(),J(1,5,0),
>
I(5),5*1000,5*100,2/3,.4,
>
arate=.,vals=.,lambda=.,.,M)
: alginfo="global","d0","moptimize"
: b_glo3=amcmc(alginfo,&lregeval(),mean(b_start),
>
variance(b_start),10000,0,.,.,
>
arate=.,vals=.,(2.38^2/5),.,M)
: arate´
.2253
: mean(b_glo3)´
1
1
2
3
4
5

-.0000916295
-.0064095109
.0054916501
40.14276799
2.497166774

12. One might wonder why I did not retain the values of lambda from the initial run and submit these
- this is because the global sampler requires a scalar value for lambda, while the metropolis-withingibbs run returns a vector of values overwritten in lambda.

: end

Apparently, the proposal distribution was tuned pretty successfully in the initial
run with the metropolis-within-gibbs sampler. The mean values of the parameters
obtained from the global draw are close to their maximum-likelihood values and
the acceptance rate is in the healthy range.
I could have also set up this problem using a structure, and that would go
something like this:
. mata:
mata (type end to exit)
: A=amcmc_init()
: amcmc_alginfo(A,("global","d0","moptimize"))
: amcmc_lnf(A,&lregeval())
: amcmc_xinit(A,J(1,5,0))
: amcmc_Vinit(A,I(5))
: amcmc_model(A,M)
: amcmc_draws(A,4000)
: amcmc_damper(A,2/3)
: amcmc_draw(A)
: end

I can now access results using the previously described amcmc results *(A) set of
commands.

4.2

Censored Quantile Regression

While the previous example demonstrated the basic principles and how one might
apply adaptive MCMC in problems of parameter estimation, the example did not
show how the methods might work when the usual maximization-based techniques
fail. Chernozukov and Hong (2003) use as an example censored quantile regression originally developed in Powell (1984) and extended in Powell (1986), which,
as Chernozukov and Hong (2003, p. 296) note provides a way to do “valid inference in Tobin-Amemiya models without distributional assumptions and with
heteroskedasticity of unknown form.” Unfortunately, the model is hard to handle
with the usual methods. The objective function is:
Ln (θ) = −

n
X

ρτ (Yi − max [ci , Xi β])

(5)

i

Where ci in (5) denotes a (left) censoring point that might be specific to the ith
observation, and ρτ (u) = (τ − (1(u < 0))u. τ ∈ (0, 1) is the quantile of interest.
Estimation using derivative-based maximization methods is problematic because
the objective function (5) has flat regions and discontinuities. While one might
do quite well with a non-derivative based optimization method such as NelderMead, one then is confronted with the problem of characterizing the parameters’
distribution and getting standard errors. For these reasons, one might opt for a

LTE/QBE estimator.
To apply amcmc() to the problem, I first program the objective function as
follows:13
. mata:
mata (type end to exit)
: void cqregeval(M,todo,b,crit,g,H) {
>
real colvector u,Xb,y,C
>
real scalar tau
>
>
Xb
=moptimize_util_xb(M,b,1)
>
y
=moptimize_util_depvar(M,1)
>
tau
=moptimize_util_userinfo(M,1)
>
C
=moptimize_util_userinfo(M,2)
>
u
=(y:-rowmax((C,Xb)))
>
crit =-colsum(u:*(tau:-(u:<0)))
> }
note: argument todo unused
note: argument g unused
note: argument H unused
: end

The following code, which sets up a model statement for use with [M-5] moptimize( ).
One can verify by following the commands with the command moptimize(M) that
this model and variations on the basic theme obtained by dropping or adding
additional variables encounter difficulties:
. webuse laborsub, clear
. gen censorpoint=0
. mata:
mata (type end to exit)
: M=moptimize_init()
: moptimize_init_evaluator(M,&cqregeval())
: moptimize_init_depvar(M,1,"whrs")
: moptimize_init_eq_indepvars(M,1,"kl6 k618 wa")
: tau=.6
: moptimize_init_userinfo(M,1,tau)
: st_view(C=.,.,"censorpoint")
: moptimize_init_userinfo(M,2,C)
: moptimize_init_evaluatortype(M,"d0")
: end

Having set up the problem in this fashion allows usage of amcmc(), where I use
the strategy of using an initial Metropolis-within-Gibbs-type algorithm, followed
by a global sampler:
. mata:
mata (type end to exit)
: alginfo="mwg","d0","moptimize"
: b_start=amcmc(alginfo,&cqregeval(),J(1,4,0),
>
I(4),5000,1000,2/3,.4,
13. Technically, one might code the objective function without summing over observations. I sum so
that the objective is compatible with Nelder-Mead in Stata, which requires a type d0 evaluator.

>
arate=.,vals=.,lambda=.,.,M)
: alginfo="global","d0","moptimize"
: b_end=amcmc(alginfo,&cqregeval(),mean(b_start),
>
variance(b_start),20000,10000,1,.234,
>
arate=.,vals=.,lambda=.,.,M)
: end

Since this application might be of more general interest, I have developed a Stata
command which is effectively a wrapper for the LTE/QBE estimation of censored
quantile regression, called mcmccqreg.14 AftThe previous snippet of code may be
executed by the Stata command:
. set seed 584937
. qui mcmccqreg whrs kl6 k618 wa, tau(.6) sampler("mwg") draws(5000) ///
>
burn(1000) dampparm(.667) arate(.4) censorvar(censorpoint)
.
.

mat binit=e(b)
mat V=e(V)

. mcmccqreg whrs kl6 k618 wa, tau(.6) sampler("global") draws(20000) ///
>
burn(10000) arate(.234) saving(lsub_draws) replace ///
>
from(binit) fromv(V)
Powell´s mcmc-estimated censored quantile regression
Observations:
250
Mean acceptance rate:
0.219
Total draws:
20000
Burn-in draws:
10000
Draws retained:
10000
whrs

Coef.

kl6
k618
wa
_cons

-1175.616
-171.2775
-29.2276
2638.366

Std. Err.
9.740436
1.568818
.6685669
31.37331

t
-120.69
-109.18
-43.72
84.10

P>|t|
0.000
0.000
0.000
0.000

[95% Conf. Interval]
-1194.709
-174.3527
-30.53813
2576.868

-1156.523
-168.2023
-27.91708
2699.864

Value of objective function:
Mean:
-89298.99
Min:
-89295.83
Max:
-89308.63
Draws saved in: lsub_draws
*Results are presented to conform with Stata covention, but
are summary statistics of draws, not coefficient estimates.

One can see from the way the command is issued how information about the
sampler, the drawing process, the censoring point (which has default of zero for
all observations), can be controlled using the mcmccqreg command. The command
produces “estimates” which are summary statistics of the sampling run. mcmccqreg
allows one to save results, and the results of the run are saved in the file lsub draws
along with the objective function value after each draw. The user can then easily
analyze the draws using Stata’s graphing and statistical analysis tools. While
the workings of the command derive more or less directly from the description of
amcmc(), more information about the command, and some additional examples,
can be found by accessing the mcmccqreg’s help file.
14. findit ssc mcmccqreg.

4.3

Drawing from a distribution

I now show how to use amcmc() to draw from a distribution. Suppose that I have
developed a theory that says three variables are jointly distributed according to a
distribution characterized by:
p(x1 , x2 , x3 ) ∝ exp −x21 − 0.5x22 + x1 x2 − 0.05(x3 − 100)2



As written, p does not integrate to one and seems hard to invert. While metropoliswithin-gibbs or global sampling works fine with this example, to illustrate the block
sampler, I will draw from the distribution in blocks, where values for the first two
arguments are drawn together, followed by a draw of the third. Thus, the block
matrix to be passed to amcmc() is:
B=



1
0

1
0

0
1



The code which programs the function and draws from the distribution is as follows:
. set seed 262728
. mata:
mata (type end to exit)
: real scalar ln_fun(x)
> {
>
return(-x[1]^2-1/2*x[2]^2+x[1]*x[2]-.05*(x[3]-100)^2)
> }
: B=(1,1,0) \ (0,0,1)
: alginfo="standalone","block"
: x_block=amcmc(alginfo,&ln_fun(),J(1,3,0),
>
I(3),4000,200,2/3,.4,
>
arate=.,vals=.,lambda=.,B)
: end

The example is set up to draw 4000 values, with a burn-in period of 200. A
graphical depiction of the simulation results are shown in figures 5 and 6: The
graphical depiction at once gives a visual idea as to what the marginal distributions
for the variables might look like, while the time series diagram verifies that our
simulation run seems to be getting good coverage and rapid convergence to the
target distribution.
A different way to draw from this distribution would be to set up an adaptive
MCMC problem via a structured set of commands:
. mata:
mata (type end to exit)
: A=amcmc_init()
: amcmc_lnf(A,&ln_fun())
: amcmc_alginfo(A,("standalone","block"))
: amcmc_draws(A,4000)
: amcmc_burn(A,200)
: amcmc_damper(A,2/3)
: amcmc_xinit(A,J(1,3,0))

Figure 5: Draws and the log-value of the distribution.

: amcmc_Vinit(A,I(3))
: amcmc_blocks(A,B)
: amcmc_draw(A)
: end

4.4

Bayesian estimation of a Mixed Logit Model

In this section, I describe estimation the nuts and bolts of Bayesian estimation
of a mixed logit model; the implementation is available via the Stata command
bayesmlogit, which I have written and made available for download online.15 The
Stata wrapper function bayesmlogit adds some bells and whistles, but essentially
works as described in this section.
While there is no strong reason to prefer using the amcmc routines as a function
or a structure in the previous examples, in this example the power and flexibility
of structured objects in Mata is indispensable. The problem is to estimate a mixed
logit model using Bayesian methods. My exposition of the basic ideas follows Train
(2009) as closely as possible, which also contains a nice overview of the principles.
The example supposes that one has access to the data set traindata.dta, which
is used by Hole (2007) to illustrate estimation of a mixed logit model by maximum
simulated likelihood.16
The data concerns n = 1, 2, 3, . . . , N people, each of whom makes a selection
from among j = 1, 2, 3, . . . , J choices on occasions t = 1, 2, 3, . . . , T . For each
choice made, there are a set of covariates xnjt that explain n’s choices at t. A
person’s utility from the jth choice on occasion t is specified as:
Unjt = βn′ xnjt + ǫnjt

(6)

where in equation (6), ǫnjt is iid extreme value, and βn are individual-specific
parameters. Variation in these parameters across the population is captured by
assuming parameters normally distributed with mean b and covariance matrix W .
Denote a person’s choice at t as ynt ∈ J. Then, the likelihood that person n
chooses j at t is:
Y eβn′ xnynt t
L(yn |β) =
(7)
PJ β ′ x
e n njt
j=1
t

Given the distribution of β, I can write the above conditional on the distribution
of parameters, φ(β|b, W ), and integrate over the distribution of parameter values
to get:
Z
L(yn |b, W ) =

L(yn |β)φ(β|b, W )dβ

(8)

In a Bayesian approach, a prior k(b, W ) is assumed, and the joint posterior likeli15. Type findit bayesmlogit from the stata prompt
16. The data is downloadable from Train’s website at: (http://elsa.berkeley.edu/ train/). The help file
for amcmc - accessible by typing either help mata amcmc() or help mf amcmc at Stata’s command
prompt - describes an example that relies on data downloadable from the Stata website.

hood of the parameters are formed using:
K(b, W |Y, X) ∝

Y

L(yn |b, W )k(b, W )

(9)

n

Because computation of the likelihood in equation (9) is difficult, simulation-based
methods are usually employed in estimation, as in the Stata package mixlogit
developed in Hole (2007)17 . An alternative is a Bayesian approach. As described
by Train (2009), estimation becomes fairly easy (at least conceptually) if one breaks
the problem into a sequence of conditional distributions, taking the view that each
set of individual-level coefficients βn are additional parameters to be estimated.
The posterior distribution of parameters given data becomes:
K(b, W, βn , n = 1, 2, 3, . . . , N |y, X) ∝

Y

L(yn |βn )φ(βn |b, W )k(b, W )

(10)

n

Following the recipe given in Train (2009, p.301-2), drawing from the posterior
proceeds in three steps. First, b is drawn conditional on βn and W ; then, W is
drawn conditional on b and βn , and finally the values of βn are drawn conditional
on b and W . The first two steps are straightforward, assuming that the prior
distribution of b is normal with extremely large variance, and that the prior for W is
an inverted Wishart with K degrees of freedom and an identity scale matrix. In this
case, the conditional distribution of b is N (β, W N −1 ), where β is the mean of the
βn ’s. The conditional distribution of W is inverted Wishart with
PK + N degrees of
freedom and scale matrix (KI +N S)/(K +N ), where S = N −1 n (βn −b)(βn −b)′
is the sample variance of the βn ’s about b.
The distribution of βn given choices, data, and (b, W ) has no simple form, but
from equation (10), we see that the distribution of a particular person’s parameters
obeys:
K(βn |b, W, yn , Xn ) ∝ L(yn |βn )φ(βn |b, W )
(11)
Where the term L(yn |βn ) in equation (11) is given by equation (7). This is a
natural place to apply MCMC methods, and it is here where I can employ the
amcmc *() suite of commands.
I now return to the example. traindata.dta contains information on the energy contract choices of 100 people, where each person faces up to 12 different
choice occasions. Suppliers’ contracts are differentiated by price, the type of contract offered, whether or not they were local to the individual, whether or not the
supplier is well-known, and the season in which the offer was made.
As a point of comparison, I first estimate the model in Train (2009, p. 305),
using mixlogit (after download and installation):
. clear all
. set more off
. use traindata.dta
. set seed 90210
. mixlogit y, rand(price contract local wknown tod seasonal) group(gid) id(pid)
Iteration 0:
log likelihood = -1253.1345 (not concave)
17. From the Stata prompt: net search mixlogit.

Iteration
Iteration
Iteration
Iteration
Iteration
Iteration

1:
2:
3:
4:
5:
6:

log
log
log
log
log
log

likelihood
likelihood
likelihood
likelihood
likelihood
likelihood

=
=
=
=
=
=

-1163.1407
-1142.7635
-1123.6896
-1122.6326
-1122.6226
-1122.6226

(not concave)

Mixed logit model

Number of obs
LR chi2(6)
Prob > chi2

Log likelihood = -1122.6226
Std. Err.

z

P>|z|

=
=
=

4780
467.53
0.0000

y

Coef.

[95% Conf. Interval]

price
contract
local
wknown
tod
seasonal

-.8908633
-.22285
1.958347
1.560163
-8.291551
-9.108944

.0616638
.0390333
.1827835
.1507413
.4995409
.5581876

-14.45
-5.71
10.71
10.35
-16.60
-16.32

0.000
0.000
0.000
0.000
0.000
0.000

-1.011722
-.2993539
1.600098
1.264715
-9.270633
-10.20297

-.7700045
-.1463462
2.316596
1.85561
-7.312469
-8.014916

price
contract
local
wknown
tod
seasonal

.1541266
.3839507
1.457113
-.8979788
1.313033
1.324614

.0200631
.0432156
.1572685
.1429141
.1648894
.1881265

7.68
8.88
9.27
-6.28
7.96
7.04

0.000
0.000
0.000
0.000
0.000
0.000

.1148036
.2992497
1.148873
-1.178085
.9898559
.9558927

.1934495
.4686516
1.765354
-.6178722
1.63621
1.693335

Mean

SD

To implement the Bayesian estimator, I proceed in the steps outlined by Train
(2009, p.301-2). First, I develop a Mata function that produces a single draw from
the conditional distribution of b:
. mata:
mata (type end to exit)
: real matrix drawb_betaW(beta,W) {
>
return(mean(beta)+rnormal(1,cols(beta),0,1)*cholesky(W)´)
> }
: end

Next, I use the recipe described in Train (2009, p.299) to draw from the conditional
distribution of W . The Mata function is:
. mata
mata (type end to exit)
: real matrix drawW_bbeta(beta,b)
> {
>
v=rnormal(cols(b)+rows(beta),cols(b),0,1)
>
S1=variance(beta)
>
S=invsym((cols(b)*I(cols(b))+rows(beta)*S1)/(cols(b)+rows(beta)))
>
L=cholesky(S)
>
R=(L*v´)*(L*v´)´/(cols(b)+rows(beta))
>
return(invsym(R))
> }
: end

I now have two of the three steps of the drawing scheme in place. The last task is
a bit more nuanced, and involves using structured amcmc problems in conjunction

with the flexible ways in which one can manipulate structures in Mata. The key
is to think of drawing each set of individual-level parameters βn as a separate
adaptive MCMC problem. It is helpful to first get all the data into Mata, get
familiar with its structure, and then work from there:
. mata:
mata (type end to exit)
: st_view(y=.,.,"y")
: st_view(X=.,.,"price contract local wknown tod seasonal")
: st_view(pid=.,.,"pid")
: st_view(gid=.,.,"gid")
: end

The matrix (really, a column vector) y is a sequence of dummy variables marking
the choices of individual n in each choice occasion, while the matrix X collects
explanatory variables for each potential choice. pid and gid are identifiers for
individuals and choice occasions, respectively. I now write a Mata function that
computes the log-probability for a particular vector of parameters for a given
person, conditional on that person’s information:
. mata:
:
>
>
>
>
>
>
>
>
>
>
:

mata (type end to exit)
real scalar lnbetan_bW(betaj,b,W,yj,Xj)
{
Uj=rowsum(Xj:*betaj)
Uj=colshape(Uj,4)
lnpj=rowsum(Uj:*colshape(yj,4)):ln(rowsum(exp(Uj)))
var=-1/2*(betaj:-b)*invsym(W)*(betaj:-b)´1/2*ln(det(W))-cols(betaj)/2*ln(2*pi())
llj=var+sum(lnpj)
return(llj)
}
end

The function takes in five arguments, the first of which is a parameter vector for
the person; the values to be drawn. The second and third arguments characterize
the mean and covariance matrix of the parameters across the population.18 The
fourth and fifth arguments contain information about an individual’s choices and
explanatory variables.
The first line of code multiplies parameters by explanatory variables to form
utility terms, which are then shaped into a matrix with four columns; in the data,
on each choice occasion individuals have four options available, so after the reshaping the utilities from potential choices on each occasion occupy a row, with separate
choice occasions in columns. lnpj then contains the log probabilities of the choices
actually made; the log of utility less the (logged) sum of exponentiated utilities.
Finally, var computes the (log) distribution of parameters about the conditional
mean, and llj sums the two components. The result is the log-likelihood of in18. In the interests of clarity this function is not as fast as it could be, and it is also specific to the
data set. One way of speeding the algorithm is to compute the Cholesky decomposition of W once
before individual-level parameters are drawn. The Stata wrapper bayesmlogit exploits this and a
few other improvements.

dividual n’s parameter values, given choices, data, and the parameters governing
the distribution of individual-level parameters.
I now set up a structured problem for each individual in the data set. I begin
by setting up a single adaptive MCMC problem, and then replicate this problem
using [M-5] J( ) to match the number of individual-level parameter sets - the same
as the number of individual-level identifiers in the data (gid) - characterized via
Mata’s [M-5] panelsetup( ) command:
. mata
mata (type end to exit)
: m=panelsetup(pid,1)
: Ap=amcmc_init()
: amcmc_damper(Ap,1)
: amcmc_alginfo(Ap,("standalone","global"))
: amcmc_append(Ap,"overwrite")
: amcmc_lnf(Ap,&lnbetan_bW())
: amcmc_draws(Ap,1)
: amcmc_append(Ap,"overwrite")
: amcmc_reeval(Ap,"reeval")
: A=J(rows(m),1,Ap)
: end

I also apply the amcmc option ‘‘overwrite’’, which means that only the results
from the last round of drawing will be saved. The specification of the "reeval"
option means that each individual’s likelihood will be reevaluated at the new parameter values and the old values of coefficients before drawing.
I now duplicate the problem in forming a matrix of adaptive MCMC problems
- one for each individual - and then use a loop to fill in individual-level choices and
explanatory variables as arguments. In the end, the “matrix” A is a collection of
100 separate adaptive MCMC problems. Prior to doing this, some initial values
for b and W are set, and some initial values for individual-level parameters are
drawn. I set up a pointer matrix Args to hold this information, along with the
individual-level information.
. mata
mata (type end to exit)
: Args=J(rows(m),4,NULL)
: b=J(1,6,0)
: W=I(6)*6
: beta=b:+sqrt(diagonal(W))´:*rnormal(rows(m),cols(b),0,1)
: for (i=1;i<=rows(m);i++) {
>
Args[i,1]=&b
>
Args[i,2]=&W
>
Args[i,3]=&panelsubmatrix(y,i,m)
>
Args[i,4]=&panelsubmatrix(X,i,m)
>
amcmc_args(A[i],Args[i,])
>
amcmc_xinit(A[i],b)
>
amcmc_Vinit(A[i],W)
>
: end

}

After creating some placeholders for the draws (bvals and Wvals), the drawing
algorithm can be executed as follows:
. mata
mata (type end to exit)
: its=20000
: burn=10000
: bvals=J(0,cols(beta),.)
: Wvals=J(0,cols(rowshape(W,1)),.)
: for (i=1;i<=its;i++) {
>
b=drawb_betaW(beta,W/rows(m))
>
W=drawW_bbeta(beta,b)
>
bvals=bvals\b
>
Wvals=Wvals\rowshape(W,1)
>
beta_old=beta
>
for (j=1;j<=rows(A);j++) {
>
amcmc_draw(A[j])
>
beta[j,]=amcmc_results_lastdraw(A[j])
>
}
> }
:
: end

The algorithm consists of an outer loop and an inner loop, within which individuallevel parameters are drawn sequentially. The current value of the beta vector,
which holds individual-level parameters in rows, is overwritten with the last draw
produced using the command amcmc results lastdraw().
A subtlety of the code also indicates a reason why it is useful to pass additional function arguments as pointers; each time a new value of b and W is drawn,
I do not need to reiterate to each sampling problem that b and W have changed,
because pointers point to positions that hold objects, not the values of the objects
themselves. Thus, every time a new value of b or W is drawn, the arguments of all
100 of the problems are automatically changed. By specifying that the target distribution for each individual level problem is to be reevaluated, the routine knows
that it needs to recalculate lnbetan bW at the last drawn value when comparing a
new draw to the previous one.
Since the technique might be of greater interest, I have developed a Stata
command implementing the algorithm called bayesmlogit.19 As an illustration,
the algorithm described by the previous code snippet could be executed with the
following command, which also summarizes results in a way conformable with
usual Stata output:
. set seed 475446
. bayesmlogit y, rand(price contract local wknown tod seasonal) ///
>
group(gid) id(pid) draws(20000) burn(10000) ///
>
samplerrand("global") saving(train_draws) replace
Bayesian Mixed Logit Model
Acceptance rates:
Fixed coefs
19. findit ssc bayesmlogit.

=

Observations
Groups
Choices
Total draws

=
=
=
=

4780
100
1195
20000

Random coefs(ave,min,max)= 0.270, 0.235, 0.289
Std. Err.

t

Burn-in draws
P>|t|

=

10000

y

Coef.

[95% Conf. Interval]

price
contract
local
wknown
tod
seasonal

-1.168711
-.3433208
2.637242
2.138963
-11.16374
-11.19243

.1245738
.0682585
.3436764
.2596608
1.049769
1.030291

-9.38
-5.03
7.67
8.24
-10.63
-10.86

0.000
0.000
0.000
0.000
0.000
0.000

-1.4129
-.4771212
1.963567
1.629976
-13.2215
-13.212

-.9245209
-.2095204
3.310917
2.647951
-9.105983
-9.172849

Cov_Random
var_price
cov_pricec~ t
cov_pricel~ l
cov_pricew~ n
cov_pricetod
cov_prices~ l
var_contract
cov_contra~ l
cov_contra~ n
cov_contra~ d
cov_contra~ l
var_local
cov_localw~ n
cov_localtod
cov_locals~ l
var_wknown
cov_wknown~ d
cov_wknown~ l
var_tod
cov_todsea~ l
var_seasonal

.8499292
.1128769
1.583028
.8898662
6.106009
6.044055
.3450904
.4714882
.3624791
.7592097
.9147682
7.000292
4.022065
12.84674
13.40598
3.364285
6.513209
7.109282
57.62449
53.93841
55.05572

.2332495
.0803203
.4519537
.3096053
1.909356
1.892895
.0670202
.2131141
.1560516
.6576296
.65939
1.883972
1.248119
3.787742
3.727253
1.012474
2.60766
2.563623
16.97876
16.35184
16.54599

3.64
1.41
3.50
2.87
3.20
3.19
5.15
2.21
2.32
1.15
1.39
3.72
3.22
3.39
3.60
3.32
2.50
2.77
3.39
3.30
3.33

0.000
0.160
0.000
0.004
0.001
0.001
0.000
0.027
0.020
0.248
0.165
0.000
0.001
0.001
0.000
0.001
0.013
0.006
0.001
0.001
0.001

.3927132
-.044567
.6971079
.2829775
2.363286
2.333601
.2137174
.0537416
.0565865
-.5298765
-.3777688
3.307328
1.575501
5.422006
6.099812
1.379632
1.401671
2.084064
24.3427
21.88551
22.62226

1.307145
.2703208
2.468948
1.496755
9.848732
9.75451
.4764634
.8892347
.6683717
2.048296
2.207305
10.69326
6.468629
20.27148
20.71214
5.348938
11.62475
12.1345
90.90628
85.99131
87.48918

Random

Draws saved in train_draws
*Results are presented to conform with Stata covention, but
are summary statistics of draws, not coefficient estimates.

The results are similar but not identical to those obtained using mixlogit. Additional information and examples about the workings of bayesmlogit can be found
in the help file, and some applications of estimation of a mixed logit model using Bayesian methods are provided in the help file for amcmc(), accessible via the
commands help mf amcmc or help mata amcmc()..

5

Description

In this section I sketch a Mata implementation of what I have been referrring to
as a global adaptive MCMC algorithm. The sketched routine omits a few details,
mainly about parsing options, but is relatively true to form in describing how the
algorithms described in the paper are actually implemented in Mata, and might
be used as a template for developing more specialized algorithms. It assumes that
the user wishes to draw from a stand-alone function without additional arguments.
The code:
. mata:
mata (type end to exit)

:
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
>
:

real matrix amcmc_global(f,xinit,Vinit,draws,burn,damper,
aopt,arate,val,lam)
{
real scalar nb,old,pro,i,alpha
real rowvector xold,xpro,mu
real matrix Accept,accept,xs,V,Vsq,Vold
nb=cols(xinit)
xold=xinit
lam=2.38^2/nb
old=(*f)(xold)
val=old

/* Initialization */

Accept=0
xs=xold
mu=xold
V=Vinit
Vold=I(cols(xold))
for (i=1;i<=draws;i++) {
accept=0
Vsq=cholesky(V)´
/* Prep V for drawing */
if (hasmissing(Vsq)) {
Vsq=cholesky(Vold)´
V=Vold
}
xpro=xold+lam*rnormal(1,nb,0,1)*Vsq

/* Draw, value calc. */

pro=(*f)(xpro)
if

(pro==. ) alpha=0

/* calculation of accept. pro

b */
else if (pro>old) alpha=1
else alpha=exp(pro-old)
if (runiform(1,1)<alpha) {
old=pro
xold=xpro
accept=1
}
lam=lam*exp(1/(i+1)^damper*(alpha-aopt)) /*update*/
xs=xs\xold
val=val\old
Accept=Accept\accept
mu=mu+1/(i+1)^damper*(xold-mu)
Vold=V
V=V+1/(i+1)^damper*((xold-mu)´(xold-mu)-V)
_makesymmetric(V)
}
val =val[burn+1::draws,]
arate=mean(Accept[burn+1::draws,])
return(xs[burn+1::draws,])
}
end

The function starts by setting up a variable (nb) to hold the dimension of the
distribution, and xold, which functions as xt in the algorithms described in table
3, is set to the user-supplied initial value. The initial value of λ (called lam) is set
as discussed by Andrieu and Thoms (2008)[ p.359].
Next, the log-value of the distribution (f) at xold is calculated and called old.
The next few steps proceed about as one would expect. However, I have found it
useful to have a default covariance matrix waiting - Vold in the code - to safeguard
for the possibility that the Cholesky decomposition might encounter problems.
This could happen if, for example, the initial variance-covariance matrix is not
positive definite, or if there is insufficient variation in the draws, which sometimes
happens in the early stages of a run. Once a useable covariance matrix has been
obtained, xpro (which functions as Yt in the algorithms in tables 1, 2, and 3) is
formed using a conformable vector of standard normal random variates, and the
function is evaluated at xpro.
The acceptance probability alpha is then calculated in a numerically stable
way in an if-else if-else block. First, if it is the case that the target function,
when evaluated, has returned a missing value, alpha is set to zero so the draw will
not be retained. Next, alpha is set to one if the proposal produces a higher value
of the target function, and otherwise, it is set as described by the algorithms.20
Finally, a uniform random variable is drawn which determines whether or not the
draw is to be accepted. Once this is known, all values are updated according
to the scheme described in table 3. Once the for loop concludes, the algorithm
overwrites the acceptance rate arate and the function value val and returns the
results of the draw.

6

Conclusions

I have given a brief overview of adaptive MCMC methods and how they might be
implemented through usage of the Mata routine amcmc() and through a suite of
commands amcmc *(). While I have given some ideas about how one might use
and display results obtained, my primary purpose is to present and describe an
implementation of adaptive MCMC algorithms. What one should do once draws
from an adaptive MCMC algorithm have been obtained has been left up to the
user. Describing and analyzing results from obtained via MCMC is the subject of
a large literature, an important part of which concerns judging when convergence
of the algorithm has been achieved. A further issue is how one should deal with autocorrelation between draws. Whatever means are employed to analyzing results,
it is fortunate that Stata provides a ready-made battery of tools for summarizing,
modifying, and graphing results.
On the subject of convergence, there does not appear to be any universally
accepted criterion, but many of guidelines have been proposed. Gelman and Rubin
(1992) present several useful ideas. A general discussion appears in Geyer (2011),
and some practical advice appears in Gelman and Shirley (2011), who advocate,
among other things, discarding the first half of a run as a burn-in period, and
20. The Mata function exp() does not evaluate to missing for very small values, as it does for very
large values.

performing multiple runs in parallel from different starting points and comparing
results. As may have been clear from the examples presented in section 4, another
option is to run the algorithm for some suitable amount of time, and then restart
the run without adaptation using previous results as starting values so that one is
drawing from an invariant proposal distribution. A perhaps overly simplistic yet
useful starting point in judging convergence is seeing whether or not the algorithm
produces results whose graphs looks like those in figure 2 - but not those in figure
4. If the graph doesn’t contain jumps or flat spots, and looks more or less like
white noise, this is a preliminary indication that the algorithm is working well.
But the fact remains that pseudo-convergence can be very difficult to detect. In
addition to containing much practical advice, Geyer (2011) also offers that one
should at do an overnight run, adding only half in jest that “...one should start
a run when the paper is submitted and keep running until the referee’s reports
arrive. This cannot delay the paper, and may detect pseudo-convergence.”(Geyer
2011, p. 18)
On the second topic, one approach is to investigate the autocorrelation function
of results and then “thin” the results, retaining only a fraction of the draws,
so that most of the autocorrelation is rid from the data. A further possibility,
discussed by Gelman and Shirley (2011), is to jumble the results of the simulation.
A very good place to start with these and other aspects of analyzing results is
Brooks, Gelman, Jones, and Meng (2011).

7

